


EJERCICIO 5 – Timer TMR0-

CONEXIONES DE LA PLACA BASICA DE APENDIZAJE:

Para este ejercicio necesitamos la placa básica de aprendizaje, en ella tenemos conectado a las líneas RA0..RA3, 4 diodos led y a las líneas RB5, RB4 y RB0, tres pulsadores de la siguiente manera:



Como vamos a utilizar la entrada del TMR0 como contador de eventos, tendremos que desconectar la resistencia del pulsador conectado a RB5 y conectarla a la entrada del timer RA4/T0CKI.

EJERCICIO 5-0:

El objetivo de este ejercicio es construir un contador de eventos con el timer TMR0, de tal manera que cada vez que este pase de 255 a 0, incrementará un contador hecho con el PORTA, cuyo valor binario se visualizará en los leds.

El TMR0 del PIC16F84:

Para manejar la TMR0, el PIC utiliza 2 registros, estos son:

- **TMR0 (01h):** Este es el contador propiamente dicho y puede ser leído y escrito en cualquier momento.
- **OPTION (81h):** De este registro, utilizaremos los siguientes bits:

Bit 7	Bit 6	Bit 5	Bit 4	Bit 3	Bit 2	Bit 1	Bit 0
-	-	T0CS	T0SE	PSA	PS2	PS1	PS0

- **T0CS:** Indica cual es el origen de los pulsos de clk para el timer; si es 1 el origen es la señal introducida por el pin RA4/T0CKI, si es 0 el origen será el ciclo de instrucción ($CLKOUT = FrecuenciaXtal/4$).
- **T0SE:** Indica el flanco por el que se incrementará la cuenta del contador TMR0. Si vale 0 se hará por el flanco de bajada, si es 1 se hará por el flanco de subida.
- **PSA:** El chip tienen internamente un divisor, este se puede asignar mediante este bit al WDT con un 1 o al TMR0 con un 0.
- **PS2:PS0:** Este es el valor de la preescala del divisor, según la siguiente tabla de valores.

Valor PS2:PS0	TMR0	WDT
000	1:2	1:1

001	1:4	1:2
010	1:8	1:4
011	1:16	1:8
100	1:32	1:16
101	1:64	1:32
110	1:128	1:64
111	1:256	1:128

El TMR0 puede funcionar de dos maneras; como contador de eventos o como temporizador. Para seleccionar de que manera queremos que funciones utilizamos el bit T0CS del registro OPTION_REG.

Cuando seleccionamos TMR0 como contador de eventos, mediante el pin T0SE indicamos por que flanco, de subida o bajada, en la señal aplicada en el pin RA4/T0CKI, queremos que se incremente el contador.

La señal que lleva al TMR0 la podemos preescalar mediante un divisor, este divisor lo podemos asignar tanto al WDT como al TMR0 mediante el bit PSA del OPTION_REG. El valor del rango del divisor viene determinado por los bit PS2:PS0 según la tabla anterior.

El TMR0 es capaz de realizar una interrupción cada vez que pasa de 255 a 0, para permitirla hay que poner a 1 el bit T0IE del registro INTCON y el flag de la interrupción es el T0IF del registro INTCON.

Pasemos a ver el Ejer5-0.asm, cuyo fuente es el siguiente:

```

Include p16f84.inc

; CONSTANTES

ValIni equ 250           ;Valor inicial del timer

;PROGRAMA

org H'0000
goto Inicio

org H'0004               ;VECTOR DE INTERRUPCION
btfsc INTCON,T0IF      ;Comprueba si la int. es del timer
goto IntExt
retfie

IntExt:
incf PORTA
movlw ValIni           ;valor inicial del timer 0.
movwf TMR0
bcf INTCON,T0IF
retfie

Inicio:
Bsf STATUS,RP0        ;Activa la pagina 1
Movlw b'11110000      ;RA0..RA3 como salida.
Movwf TRISA
movlw b'11111000      ;Valor de la preescala.
movwf OPTION_REG
Bcf STATUS,RP0        ;Activa la pagina 0

```

```

movlw H'00
Movwf PORTA
movlw ValIni           ;valor inicial del timer 0.
movwf TMR0

bsf INTCON,GIE        ;Activa todas las interrupciones.
bsf INTCON,T0IE      ;Activa la interrupcion del timer.

Fin:      Goto fin           ;Bucle de parada.

End

```

Analizamos el programa:

El valor de TMR0 se inicializa con ValIni (250), con lo que cada 6 pulsaciones en la tecla conectada al pin RA4/T0CKI, el TMR0 pasara de 255 a 0, generando una interrupción que incrementara el valor del contador formado por los leds conectados al PORTA.

Lo único a destacar es la programación del OPTION_REG para configurar el TMRO. Veamos dos ejemplos:

- OPTION_REG = b'11111000 : Con este valor hacemos T0CS=1 (pone el timer en modo contador de eventos), T0SE=1 (el contador se incrementará con el flanco de bajada de la señal el RA4/T0CKI), PSA=1 (el divisor esta asignado al WDT por lo que el ratio de la señal será 1:1). Con esta configuración se necesitarán 6 pulsos para incrementar el valor de los leds.
- OPTION_REG = b'11110000 : Con este valor hacemos T0CS=1 (pone el timer en modo contador de eventos), T0SE=1 (el contador se incrementará con el flanco de bajada de la señal el RA4/T0CKI), PSA=0 (el divisor esta asignado al TMR0 por lo que el ratio de la señal vendrá determinado por el valor de PS2:PS0, como es 000 el rango será 1:2). Con esta configuración se necesitarán 12 pulsos para incrementar el valor de los leds.

Puedes probar diferentes valores de OPTION_REG, para estudiar más en profundidad el funcionamiento del TMR0.

Si utilizamos el TMR0, junto con el divisor, como temporizador podremos medir periodos de tiempo muy precisos, hacer periodos de espera o generar señales de tipo PWM.